
Вступление

Если содержательно представить себе поведение
известных триггеров и регистров, то оно будет выглядеть примерно так:

всегда, когда <что-то> происходит,
значения на выходах <как-то> изменяются

Синхронный триггер/регистр:
I <что-то>: передний фронт тактового сигнала
I <как-то>: как записано в таблице значений

Асинхронный триггер/регистр:
I <что-то>: изменение значений на входах
I <как-то>: сохраняются или изменяются согласно таблице значений

Такое описание поведения асинхронных регистров
подходит и для комбинационных схем

2 / 18



V: реагирующая процедура
всегда, когда <что-то> происходит,

значения на выходах <как-то> изменяются

В V в программной семантике:
I “всегда” = “процедура always”
I “<что-то> происходит” = “выполняется событие (<что-то>)”
I “значение изменяется” =

“блокирующее/неблокирующее присваивание”1

Процесс, имеющий такое поведение, записывается следующим образом:2

always @(<список событий>) <команда>

Для ясности будем называть такой процесс реагирующей процедурой

1 Есть и другие виды присваиваний, но можно обойтись и без них
2 Более точно, <список событий> относится не к процессу, а к <команде> —
но в V поддерживается только такая форма записи, на ней и остановимся

V: реагирующая процедура
always @(<список событий>) <команда>

Семантика: <команда> выполняется
после каждого выполнения любого <события> из <списка>

Основные виды <событий>:
I <имя точки>: событие смены значения,

“значение в точке изменилось”
I posedge <имя одноразрядной точки>: событие переднего фронта,

“в точке наступил передний фронт”
I negedge <имя одноразрядной точки>: событие заднего фронта,

“в точке наступил задний фронт”

<События> в <списке> разделяются запятой или словом or

V: реагирующая процедура

Пример:
always @(x, posedge y or negedge z) b = a;

Как это работает: присваивание “a = b” выполняется каждый раз,
когда в системе происходит хотя бы одно из трёх событий:

I изменяется значение в точке x
I наступает передний фронт в одноразрядной точке y
I наступает задний фронт в одноразрядной точке z

V: always и синхронные регистры

Пример: поддерживаемое описание D-триггера
reg q;
always @(posedge clk) q <= d;

Общие требования к поддерживаемому описанию
синхронного регистра (как реагирующей процедуры)1

1. Список событий: ровно одно событие фронта
2. Все присваивания процедуры неблокирующие
3. Выходы регистра: все переменные в левых частях присваиваний
4. Входы регистра:

I тактовый: указанный в списке событий
I остальные: точки, значения которых используются в процедуре

1 Также необходимо соблюдать
все требования поддержки отдельных процедурных команд

V: always и синхронные регистры
Чуть более нетривиальный пример:

reg [1:0] x;
reg y;
always @(posedge clk) begin

if(& a) begin
x <= 2;
y <= 1;

end
else if(!b) x <= a;

end

V: always и синхронные регистры
Чуть более нетривиальный пример:

reg [1:0] x;
reg y;
always @(posedge clk) begin

if(& a) begin
x <= 2;
y <= 1;

end
else if(!b) x <= a;

end

Таблица значений регистра:

V: always и синхронные регистры
Пример: поддерживаемое описание D-триггера с асинхронным сбросом

reg q;
always @(posedge clk, posedge rst)

if(rst) q <= 0;
else q <= d;

Общие требования к поддерживаемому описанию
синхронного регистра с дополнительными асинхронными входами

1. Список событий: непустой набор событий фронтов

2. Команда процедуры:
if(<усл1>) <команда> else if(<усл2>) <команда>

else ... if(<услk>) <команда> else <команда>,
где <усл1>...<услk> — выражения,
соответствующие (7→) всем событиям списка, кроме одного:

I posedge x 7→ x
I negedge x 7→ !x или ~x

3. Все присваивания процедуры неблокирующие

V: always и синхронные регистры
Пример: поддерживаемое описание D-триггера с асинхронным сбросом

reg q;
always @(posedge clk, posedge rst)

if(rst) q <= 0;
else q <= d;

Общие требования к поддерживаемому описанию
синхронного регистра с дополнительными асинхронными входами

4. Выходы регистра: переменные в левых частях присваиваний

5. Входы регистра:
I тактовый: указанный в списке событий

и не имеющий соответствующего выражения в цепочке ветвлений
I дополнительные асинхронные: остальные указанные в списке событий
I остальные: точки, значения которых используются в процедуре

V: always и синхронные регистры
Расхождение между программной семантикой процедуры
и соответствующим синхронным регистром
с дополнительными асинхронными входами:

I Программная семантика:
I порядок условий в ветвлениях = приоритеты асинхронных входов
I асинхронные входы “срабатывают”

только в моменты фронтов согласно списку условий
I Аппаратная семантика:

I если несколько условий, соответствующих асинхронным входам,
выполняются одновременно, то поведение не специфицировано

I асинхронные входы “работают” всегда
от “включающего” фронта до “выключающего”

Компромисс:
I В программной симуляции

не “включать” несколько асинхронных входов одновременно
I В командах, описывающих действие асинхронных входов,

присваивать только константные выражения

V: always и синхронные регистры

Чуть более нетривиальный пример:
reg [1:0] x;
reg y;
always @(posedge clk, posedge a1, negedge a2)

if(!a2) begin x <= 3; y <= 0; end
else if(a1) y <= 1;
else begin x <= b; y <= x[0]; end

V: always и асинхронные регистры
Пример: поддерживаемые описания RS-триггера

reg q;
always @(s, r) if(s) q <= 1; else if(r) q <= 0;
always @(s, r) begin if(s) q <= 1; if(r) q <= 0; end

Общие требования к поддерживаемому описанию
асинхронного регистра

1. Список событий: непустой набор событий смены значения
2. В списке событий перечислены все точки,

значения которых используются в процедуре
3. Все присваивания процедуры неблокирующие
4. Выходы регистра: переменные в левых частях присваиваний
5. Входы регистра: точки, значения которых используются в процедуре
6. Для каждого выхода должен существовать

набор значений на входах, для которого
не выполняется ни одно присваивание значения в этот выход

V: always и комбинационные схемы
Пример: поддерживаемое описание мультиплексора

reg out;
always @(in0, in1, s) begin

out = 1’bx;
case(s)

1’d0: out = in0;
1’d1: out = in1;

endcase
end

Общие требования к поддерживаемому описанию
комбинационной схемы

1. Список событий: непустой набор событий смены значения

2. Для каждого разряда каждой переменной,
встречающейся в левых частях присваиваний

I либо все присваивания в этот разряд блокирующие,
I либо все присваивания в этот разряд неблокирующие

V: always и комбинационные схемы
Пример: поддерживаемое описание мультиплексора

reg out;
always @(in0, in1, s) begin

out = 1’bx;
case(s)

1’d0: out = in0;
1’d1: out = in1;

endcase
end

Общие требования к поддерживаемому описанию
комбинационной схемы

3. Выходы схемы: переменные в левых частях присваиваний

4. Входы схемы: точки списка событий, значения которых на момент
начала выполнения процедуры используются в присваиваниях

5. В списке событий должны быть перечислены все входы схемы

6. Для каждого набора значений на входах должно выполняться
хотя бы одно присваивание значения в каждый выход

V: always и комбинационные схемы
Чуть более нетривиальный пример:

wire [1:0] z;
wire u;
reg x, y;
always @(z, u) begin

x = 1’bx; y = 0;
case(z)

2’d0: x = u;
2’d3: begin y = !u; x = y; end

endcase
end

Аппаратная семантика — комбинационная схема
со входами u (ширины 1), z (ширины 2) и выходами x, y (ширины 1),
реализующая систему функций со следующей таблицей значений:

u

z

0 00 -0 -0 11
1 10 -0 -0 00

V: краткое задание списка событий

Список событий поддерживаемой реагирующей процедуры, задающей
асинхронный регистр или комбинационную схему, можно устроить так:
перечислить в точности все точки, встречающиеся в правых частях
присваиваний и условиях ветвлений и команд выбора

Такое перечисление точек вручную — процесс, неустойчивый к ошибкам:
I описали процедуру, проглядели точку в правой части

и не включили её в список ⇒ не поддерживается
I переделали процедуру, добавив зависимость от новой точки,

и не обновили список ⇒ не поддерживается

Рекомендуемый способ записи реагирующей процедуры,
позволяющий избежать таких ошибок:

always @(*) <команда> или always @* <команда>

@(*) и @* = “список событий смены значения для всех точек,
встречающихся в правых частях присваиваний и в условиях <команды>”

V: краткое задание списка событий
Пример:

always @* begin
x = 1’bx; y = 0;
case(z)
2’d0: x = u;
2’d3: begin
y = !u;
x = y;

end
endcase

end

=

always @(z, u, y) begin
x = 1’bx; y = 0;
case(z)
2’d0: x = u;
2’d3: begin
y = !u;
x = y;

end
endcase

end

Небольшая тонкость

Переменная y является выходом и при этом включена в список событий

В этом нет ничего страшного: значение y на момент начала
выполнения процедуры не используется ни в одном присваивании,
а значит, не посылается на вход соответствующей схемы

